
6 PBE. MODELOS EN MATEMÁTICAS

Esta unidad tiene como objetivo continuar con la resolución de problemas a
través de la modelización y simulación siguiendo el patrón PBE (P systems By Ex-
ample) empleado en la unidad anterior, ilustrando en esta ocasión la aplicación
a problemas relevantes en Matemáticas (posibilidad de colorear un grafo con 3
colores), haciendo uso sistemas P que trabajan a modo de tejidos.

1 Ejemplo 4. Coloración de grafos

Este ejemplo trata de presentar un problema clásico NP-completo conocido
como 3-COL, es decir, la coloración de un grafo a través de 3 colores.

1. 1 Escenario

El problema 3-COL se formula como sigue: “dado un grafo no dirigido, deter-
minar si existe una coloración válida del grafo con tres colores”. Recordemos que una
coloración de un grafo no dirigido G = (V,E) con tres colores es una aplicación
f : V → {1, 2, 3} y se dirá que una tal coloración es válida si satisface la sigu-
iente condición adicional: para cada arista {u, v} ∈ E del grafo se tiene que
f(u) 6= f(v); es decir, una coloración válida asigna colores distintos a nodos del
grafo que sean adyacentes.

Por cuestiones de simplicidad de notación en la solución que se va a presen-
tar, usaremos los elementos del conjunto {R,G,B} para designar los tres colores
{1, 2, 3}. Emplearemos justamente los colores rojo (Red), verde (Green) y azul
(Blue), si bien podrı́an tomarse otros cualesquiera. El problema original es de
hecho un problema de decisión, que debe respondernos simplemente si el grafo
dado admite o no la 3-coloración.

Veamos por ejemplo el grafo de la Figura 3.
Si nos fijamos en el nodo 1, éste está conectado con 2 y 3, que a su vez están

conectados. Por tanto, los colores de los tres nodos deberán ser diferentes. Ahora

FIGURA 1: 3-COL. Instancia a resolver.

bien, el nodo 4 no está conectado con el nodo 1, luego estos dos nodos sı́ podrán
compartir color.

A continuación vamos a presentar una solución eficiente del problema 3-COL
a través de una familia de sistemas P de tejidos reconocedores con división celu-
lar.

1. 2 Modelo

Para cada m,n ∈ N, consideramos el sistema P de tejidos reconocedor con
división celular Π(〈n,m〉) = (Γ, E ,Σ,M1,M2,R, iin) definido como sigue:

El alfabeto de trabajo Γ es el conjunto

{Ai, Ri, Gi, Bi, Ti, Ri, GiBi, : 1 ≤ i ≤ n} ∪
{ai : 1 ≤ i ≤ 2n+m+ dlogme+ 11} ∪ {ci : 1 ≤ i ≤ 2n+ 1} ∪
{di : 1 ≤ i ≤ dlogme+ 1} ∪ {zi : 2 ≤ i ≤ m+ dlogme+ 6} ∪
{Aij , Pij , P ij , Rij , Gij , Bij : 1 ≤ i < j ≤ n} ∪ {b,D,D, e, T, S,N, [, yes, no}

El alfabeto del entorno es E = Γ− {yes, no}

El alfabeto de entrada Σ es el conjunto {Aij : 1 ≤ i < j ≤ n}

Los multiconjuntos iniciales contenidos en las células son:M1 = {a1, b, c1, yes, no}
yM2(n) = {D,A1, . . . , An}

El conjuntoR consta de las siguientes reglas:

(a) {[Ai]2 → [Ri]2[Ti]2, [Ti]2 → [Gi]2[Bi]2 : 1 ≤ i ≤ n}
(b) {(1, ai/ai+1, 0) : 1 ≤ i ≤ 2n+m+ dlogme+ 10}
(c) {(1, ci/c2i+1, 0) : 1 ≤ i ≤ 2n}
(d) (1, c2n+1/D, 2)

(e) (2, c2n+1/d1D, 0)

(f) {(2, di/d2i+1, 0) : 1 ≤ i ≤ dlogme}

(g) (2, D/e z2, 0)

(h) {(2, zi/zi+1, 0) : 1 ≤ i ≤ m+ dlogme+ 5}

(i) {(2, ddlogme+1Aij/Pij , 0) : 1 ≤ i < j ≤ n}

(j) {(2, Pij/RijP ij , 0) : 1 ≤ i < j ≤ n}

(k) {(2, P ij/BijGij , 0) : 1 ≤ i < j ≤ n}

(l) {(2, RiRij/RiRj , 0) : 1 ≤ i < j ≤ n}

(m) {(2, BiBij/BiBj , 0) : 1 ≤ i < j ≤ n}

(n) {(2, GiGij/GiGj , 0) : 1 ≤ i < j ≤ n}

(o) {(2, RjRj/[, 0) : 1 ≤ j ≤ n}

(p) {(2, BjBj/[, 0) : 1 ≤ j ≤ n}

(q) {(2, GjGj/[, 0) : 1 ≤ j ≤ n}

(r) (2, e [/λ, 0)

(s) (2, e zm+dlogme+6/T, 0)

(t) r21 ≡ (2, T/λ, 1)

(u) (1, b T/S, 0)

(v) (1, S yes/λ, 0)

(x) (1, b a2n+m+dlogme+11/N, 0)

(y) (1, N no/λ, 0)

La célula de entrada es iin = 2.

Sea u = (V,E) un grafo no dirigido con n vértices y m aristas. Consideramos las
funciones cod(u) = {{Aij : {Ai, Aj} ∈ E ∧ 1 ≤ i < j ≤ n}} y s(u) = 〈m,n〉. La
ejecución del sistema Π(s(u)) con entrada cod(u) se estructura en cuatro fases:

Fase de generación de coloraciones: se generan todas las posibles coloraciones
del grafo mediante la aplicación sucesiva de reglas de división en la célula
2. En esta fase se pude observar con bastante claridad el no determinismo
de los sistemas P y, a la vez, la confluencia de los mismos.

Fase de pre-chequeo: esta fase permite introducir objetos Rij , Gij , Bij , para
cada arista {i, j} del grafo, en todas las células etiquetadas por 2, a fin de
analizar la validez de la coloración que cada una de ellas codifica.

Fase de chequeo: esta fase chequea en cada célula 2 si la coloración que codi-
fica es válida, para ello hará uso de los objetos Rij , Gij , Bij .

Fase de salida: el sistema proporciona la salida correspondiente en función
de lo analizado anteriormente.



4 LAS MATEMÁTICAS DE LA VIDA. P SYSTEMS BY EXAMPLE

1.0.1. Definición en P-Lingua

A continuación se va a mostrar el código del programa escrito en P-Lingua
que especifica la familia de sistemas P antes descrita. La entrada, el grafo con-
creto, dependerá de los datos introducidos por el usuario, lo cuál veremos en el
siguiente apartado.
El código es el siguiente:

@model<tissue_psystems>

def main()
{
/* tissue P system skeleton */
call tricolor_tissue(n,m);

}

def tricolor_tissue(n,m)
{

call init_cells();
call init_multisets(n);
call init_environment(n,m);
call init_rules(n,m);

}

def init_cells()
{

@mu = [[]’1 []’2]’0;
}

def init_rules(n,m)
{

/* r1,i */ [A{i}]’2 --> [R{i}]’2 [T{i}]’2 : 1<=i<=n;
/* r2,i */ [T{i}]’2 --> [G{i}]’2 [B{i}]’2 : 1<=i<=n;

/* r3,i */ [a{i}]’1 <--> [a{i+1}]’0 : 1<=i<=2*n+m+@ceil(@log(m))+10;
/* r4,i */ [c{i}]’1 <--> [c{i+1}*2]’0 : 1<=i<=2*n;
/* r5 */ [c{2*n+1}]’1 <--> [D]’2;
/* r6 */ [c{2*n+1}]’2 <--> [d{1},noD]’0;
/* r7,i */ [d{i}]’2 <--> [d{i+1}*2]’0 : 1<=i<=@ceil(@log(m));
/* r8 */ [noD]’2 <--> [e,z{2}]’0;
/* r9,i */ [z{i}]’2 <--> [z{i+1}]’0 : 2<=i<=m+@ceil(@log(m))+5;
/* r10,i,j */ [d{@ceil(@log(m))+1},A{i,j}]’2 <--> [P{i,j}]’0 : i<j<=n, 1<=i<n;
/* r11,i,j */ [P{i,j}]’2 <--> [R{i,j},noP{i,j}]’0 : i<j<=n, 1<=i<n;
/* r12,i,j */ [noP{i,j}]’2 <--> [B{i,j},G{i,j}]’0 : i<j<=n, 1<=i<n;
/* r13,i,j */ [R{i},RR{i,j}]’2 <--> [R{i},noR{j}]’0 : i<j<=n, 1<=i<n;
/* r14,i,j */ [B{i},BB{i,j}]’2 <--> [B{i},noB{j}]’0 : i<j<=n, 1<=i<n;
/* r15,i,j */ [G{i},GG{i,j}]’2 <--> [G{i},noG{j}]’0 : i<j<=n, 1<=i<n;
/* r16,j */ [noR{j},R{j}]’2 <--> [bb]’0 : 1<=j<=n;
/* r17,j */ [noB{j},B{j}]’2 <--> [bb]’0 : 1<=j<=n;
/* r18,j */ [noG{j},G{j}]’2 <--> [bb]’0 : 1<=j<=n;
/* r19 */ [e,bb]’2 <--> []’0;
/* r20 */ [e,z{m+@ceil(@log(m))+6}]’2 <--> [T]’0;
/* r21 */ [T]’2 <--> []’1;
/* r22 */ [b,T]’1 <--> [S]’0;
/* r23 */ [S,yes]’1 <--> []’0;
/* r24 */ [b,a{2*n+m+@ceil(@log(m))+11}]’1 <--> [N]’0;
/* r25 */ [N,no]’1 <--> []’0;

}

def init_multisets(n)
{

@ms(1) = a{1},b,c{1},yes,no;
@ms(2) = D;
@ms(2) += A{i} : 1<=i<=n;

@ms(2) += A{e{i,1},e{i,2}} : 1<=i<=ne;
}

def init_environment(n,m)
{

@ms(0) = A{i},R{i},G{i},B{i},T{i},noR{i},noG{i},noB{i} : 1<=i<=n;
@ms(0) += a{i} : 1<=i<=2*n+m+@ceil(@log(m))+11;
@ms(0) += c{i} : 1<=i<=2*n+1;
@ms(0) += d{i} : 1<=i<=@ceil(@log(m))+1;
@ms(0) += z{i} : 2<=i<=m+@ceil(@log(m))+6;
@ms(0) += A{i,j},P{i,j},noP{i,j},RR{i,j},GG{i,j},BB{i,j} : i<j<=n,1<=i<n;
@ms(0) += b,D,noD,e,T,S,N,bb;

}

1. 3 Simulación

El sistema P descrito en el apartado anterior resuelve el problema 3-COL para
la fórmula indicada. P-Lingua acepta el fichero anterior y dispone de simuladores
preparados para su simulación. Ahora bien, como en el caso de la unidad ante-
rior, el archivo contiene una serie de parámetros cuyos valores dependerán de la
instancia que se quiera resolver.

Para llevar a cabo la simulación planteada en este ejemplo debemos hacer lo
siguiente:

1. Como en el ejemplo de SAT de la unidad anterior, es necesario definir una
aplicación basada en MeCoSim adaptada a este problema (en este caso
3-COL). Recordemos que la personalización se lleva a cabo empleando un
fichero de configuración Excel con las entradas y salidas (tabuladas y gráfi-
cas) deseadas y los parámetros para P-Lingua. El archivo Excel correspon-
diente está disponible como en la unidad anterior en el repositorio de apps
“http://www.p-lingua.org/mecosim/matvida/apps/”, de modo que se-
leccione este repositorio y seleccione la aplicación “Ejemplo 4 - 3-COL”
para instalarse con doble click y salimos del repositorio de apps. Añadi-
mos la aplicación con el botón “New”, y desde ahı́ seleccionar el archivo
“Ejemplo4 3col.xls”, lo que hará que en nuestro listado aparezca en primer
lugar la aplicación “3-COL”, lista para ser lanzada.

2. Acudir al gestor de repositorios para importar el archivo de modelo
(“Ejemplo4 3col.pli”) y el de escenario (“Ejemplo4 3col.ec2”), para poder
cargar el sistema P correspondiente expresado en lenguaje P-Lingua y apli-
carlo sobre el grafo indicado.

3. Lance la aplicación 3-COL mediante “Run”. Observe la apariencia de la
aplicación adaptada para 3-COL, que contiene:

Entrada para parámetros generales: el número de nodos, n, y el número
de aristas, m.
Entrada para las aristas, con dos columnas indicando los dos nodos
que conecta cada arista.

Salida detallada con los objetos presentes en cada membrana en cada
paso.

Salida “Result”, que deberá indicar si el grafo admite una 3-coloración
(“yes”) o no (“No”).

4. Cargue el modelo mediante Model > Set Model, abra el escenario descar-
gado mediante Scenario > Open, y proceda a simular mediante Simulation
> Simulate!. ¿Cuál es el resultado? ¿se admite la 3-coloración? ¿cuál es?. A
continuación puede salir de la aplicación y volver al listado de aplicaciones
de MeCoSim.

5. ¿Le ha costado responder la primera pregunta anterior? ¿Y a la segunda? Lo
más probable es que haya encontrado fácil respuesta a la primera pregunta
a través de alguna de las salidas proporcionadas, pero que le haya resultado
más difı́cil responder a la segunda cuestión. Esto se debe a que el proble-
ma a resolver no pretendı́a hallar la coloración sino resolver el problema
de decisión de determinar si el grafo puede ser coloreado o no emplean-
do 3 colores. No obstante, podemos estar interesado en este u otros muchos
problemas de grafos. Para mostrar información vsual sobre árboles y grafos
se ha desarrollado un plugin denominado “Graphs plugin”, que podemos
instalar desde el gestor de repositorios de MeCoSim> Plugins (recordemos que
tras instalarlo debemos salir del programa completo, y configurar el plugin
para que aparezca en nuestro listado de plugins introduciendo las siguien-
tes lı́neas en la ruta MeCoSim > prop > plugins-properties:

plugin-graphs = mecosim.plugins.graphsPlugin.GraphViewer
pluginname-graphs = Graph Viewer
pluginmethod-graphs = entryPoint
pluginorder-graphs = 6
pluginjar-graphs-1 = GraphsPlugin.jar
pluginparam-graphs-1 = numver
pluginparam-graphs-2 = graph
pluginparam-graphs-3 = twocolor

Este plugin permite visualizar información sobre grafos a partir de infor-
mación numérica o textual generada por el mecanismo de generación de
parámetros de MeCoSim. En este curso se empleará para visualizar tanto
el grafo inicial sin colorear como los grafos codificados al final de la com-
putación en cada membrana etiquetada con 2, conteniendo objetos de tipo
R, G o B. Ya estamos listos para arrancar de nuevo MeCoSim, y entrar en
la aplicación 3-COL. Una vez en ella, podemos probar el ejemplo anterior,
como se muestra en la Figura 2.

Hasta aquı́ el detalle de la simulación. En el próximo apartado se detallan los
resultados esperados y el análisis de los mismos.

1. 4 Resultados

FIGURA 2: 3-COL - Selección de parámetros en Graphs plugin para MeCoSim.

La idea de esta unidad es trabajar con un problema NP-completo resuelto con
un sistema P de tejidos con reglas de división celular, en este caso el problema
3-COL, además de ilustrar el mecanismo de personalización de aplicaciones de
MeCoSim para acercar las aplicaciones al usuario final.

En concreto, se va a considerar como ejemplo para instanciar el problema
3-COL a resolver el grafo de la Figura 3.

FIGURA 3: 3-COL. Instancia a resolver.

Deberá seguir los pasos descritos en el apartado de simulación hasta simular
el sistema P introducido para el grafo del ejemplo, analizando la salida del sis-
tema y empleando el plugin de visualización de grafos para mostrar tanto el grafo
inicial (dejando las opciones por defecto en la ventana de selección de parámetros
de la Figura 2, como los grafos finales obtenidos en cada membrana que codifica
una combinación de colores para los nodos del grafo (modificando en la ventana
de selección el tipo de grafo, cambiándolo por la opción del desplegable denom-
inada “Generate tree of graphs”). Esta última opción muestra la información de
forma jerárquica, de forma que seleccionemos una membrana del sistema, y un
paso (en este caso tenemos establecida en la configuración de esta aplicación que
solamente muestre el último paso de la computación) y nos muestre el grafo cod-
ificado en la membrana correspondiente. Esto es posible gracias al efecto combi-
nado del mecanismo de generación de parámetros de MeCoSim y del plugin de
visualización de grafos.

La Figura 4 muestra el grafo coloreado contenido en una de las membranas.

FIGURA 4: 3-COL. Instancia resuelta.

Se propone al alumno buscar un mı́nimo de 5 ejemplos de grafos que con-
tengan entre 4 y 8 nodos (para simplificar tanto su comprensión de la evolución
del sistema como su computación) y repita el proceso, analizando también me-
diante el mecanismo de depuración cómo se van disparando las reglas correspon-
dientes a cada una de las fases descritas en el apartado de “Modelo”.


